

ControllerSpielwiese

Online-Tummelplatz aller Controlling-Fans

5. November 2010

Sondernewsletter

Liebe Controller-Kolleginnen und -Kollegen und Leser/innen des Newsletters,

mit diesem Sonder-Newsletter empfehlen wir Ihrer Aufmerksamkeit die folgenden Fachbeiträge und Praxistipps unseres Partners

EXCELLENTControlling.

Mit freundlichen Grüßen

Ihre Redaktion

Inhalt des Newsletters:

- **Excel im Personalcontrolling - ein Überblick**
- **Autoren-Team**
- **Intensivseminar "Controlling & Finance mit Microsoft Excel"**
- **Praxisbeispiel 1: Monatliche Entgeltabrechnung mit Zuschlägen**
- **Praxisbeispiel 2: Arbeitsanfallanalyse**
- **Excel-Tipps für Personalcontroller**
- **XING-Gruppe "Excel im Controlling"**
- **Kontaktdaten**

Excel im Personalcontrolling – ein Überblick

Für Personalcontroller und Sachbearbeiter im Personalwesen ist Excel ein unverzichtbares Werkzeug zur Aufbereitung von Mitarbeiterdaten. ERP-Systemen wie SAP HR/HCM übernehmen zwar zuverlässig Lohn- und Gehaltsabrechnung, An- und Abwesenheitsverwaltung oder Kapazitätsplanungen, aber wo die „Großen“ überdimensioniert und unflexibel sind, ist Excel unschlagbar.

Wir zeigen Ihnen in diesem Newsletter Praxislösungen mit besonders nützlichen Spezialtechniken zur Auswertung von Personaldaten, zum Beispiel die Kombination aus SUMME() und WENN() in einer Matrixformel und den Einsatz von Formularelementen für dynamische Listen. Dazu gibt es wertvolle Tipps und Tricks für Personalcontroller.

Autoren-Team



Ignatz Schels



Prof. Dr.
Uwe M. Seidel

Ignatz Schels, Technik-Informatiker und Excel-Experte der ersten Stunde, leitet Spezialseminare für Excel im Controlling und schreibt Fachbücher zu Excel, Access und Windows.

Dr. Uwe M. Seidel ist Professor für Betriebswirtschaft an der Hochschule Regensburg und Berater für Unternehmen und öffentliche Institutionen auf den Gebieten Rechnungswesen, Controlling und Projektmanagement.

Das Autoren- und Referententeam Schels/Seidel verfasst unter dem Label **EXCELLENTControlling** Bücher, Fachartikel, Videos und Webcasts und bietet Seminare für fortgeschrittene Excel-Anwender an, z.B. „Controlling & Finance mit Microsoft Excel“ oder „Excel im Personalwesen“.

www.excellent-controlling.de

Die Beispiele, Tipps und Tricks stammen aus dem Buch **Das große Excel-Handbuch für Controller** (Markt & Technik-Verlag, zahlreiche Beispiele und Lösungen, für Excel 2003, 2007 und 2010. Komplette in Farbe!).



www.excellent-controlling.de

Intensivseminar “Controlling & Finance mit Microsoft Excel”

Wollen Sie noch mehr nützliche Tipps und Tricks zu Excel? Suchen Sie die besten Excel-Techniken für Controller? Brauchen Sie professionelle Lösungen für Ihre tägliche Arbeit mit Excel? Unser Seminar beantwortet Ihre Fragen.

Controlling & Finance mit Microsoft Excel.

29. November 2010 – 1. Dezember 2010

Ort: Pfaffenhofen/Ilm, Hotel Moosburger Hof

Zwei Tage **Top-Excel-Techniken**, Funktionen, dynamische Bereiche, Matrixformeln, zeitsparendes Reporting und modernes Diagrammdesign. Dazu am 3. Tag der Workshop **VBA-**

Makroprogrammierung für Controller (separat buchbar). Nutzen Sie die Gelegenheit, lassen Sie sich dieses außergewöhnliche Seminar nicht entgehen.

Gratis für jeden Teilnehmer:



Hier der Seminarprospekt mit Fax-Anmeldeformular:

<http://www.excellent-controlling.de>

Oder schreiben Sie uns einfach: info@excellent-controlling.de

Praxisbeispiel 1: Monatliche Entgeltabrechnung mit Zuschlägen

Die Aufgabe besteht darin, eine Vorlage zur Erfassung von Arbeitsstunden für Mitarbeiter zu schaffen, die automatisch die Datumsreihe für den eingestellten Monat berechnet und Wochenendtage, Feiertage und andere arbeitsfreie Tage kennzeichnet. Die Auswertung dieser „Lohnzettel“ sollte bei der Berechnung der Arbeitnehmerbezüge Zuschläge für die Wochenendstunden berücksichtigen.

Basisdaten

Das Tabellenblatt *Basisdaten* enthält das Abrechnungsjahr, eine Monatsreihe von Januar bis Dezember, den Grundlohn und die Zuschläge für Samstag und Sonntag. Alle Daten werden mit globalen Bereichsnamen versehen (Excel bis Version 2003: *Einfügen / Namen definieren*, Excel 2007/2010: *Formel / Definierte Namen / Namens-Manager*).

	A	B	C	D
1	Abrechnungsjahr:	2010		Januar
2				Februar
3	Grundlohn:	25,00 €		März
4	Zuschlag Samstag:	50%		April
5	Zuschlag Sonntag:	100%		Mai
6				Juni
7			ZSonntag	Juli
8			ZSamstag	August
9				September
10			Grundlohn	Oktober
11				November
12				Dezember
13				
14				
15				
16				

The table is titled 'Basisdaten' in the bottom right corner. Red arrows point from the labels 'ZSonntag', 'ZSamstag', and 'Grundlohn' to their respective cells in the spreadsheet.

Abbildung 1: Basisdaten mit Bereichsnamen

Lohnzettel-Vorlage mit Steuerelement

Im Tabellenblatt *Lohnzettel Vorlage* wird das aktuelle Jahr aus den Basisdaten per Verknüpfung in die Zelle B2 geholt:

B2: =Jahr

Für die Auswahl des Abrechnungsmonats zeichnen Sie ein Formularelement (Kombinationsfeld): Excel bis Version 2003: *Ansicht / Symbolleisten / Formular*, Excel 2007/2010: *Entwicklertools / Steuerelemente / Einfügen*. Die Registerkarte *Entwicklertools* muss vorher in den Excel-Optionen aktiviert werden (2007: Allgemein, 2010: Menüband anpassen).

Zum Zeichnen klicken Sie auf das Werkzeug und ziehen ein Rechteck im Tabellenblatt auf. Im Kontextmenü der rechten Maustaste finden Sie den Befehl *Steuerelement formatieren*. Geben Sie ein:

Eingabebereich: Monate

Zellverknüpfung: \$B\$3

Dropdownzeilen: 12



Abbildung 2: Lohnzettel-Vorlage mit Steuerelement

Ein Klick in eine beliebige Zelle aktiviert das Kombinationsfeld, die Ausgabeverknüpfung (\$B\$3) enthält anschließend die Nummer des gewählten Monats. Mit der Funktion DATUM() wird in der Zelle B6 der erste Tag der Datumsreihe berechnet:

=DATUM(Jahr;Monat;Tag)

Jahr und Monat holen Sie aus A1 und A2, für das Argument Tag verwenden Sie die Spaltennummer. So kann die Formel bis zum Monatsende nach rechts kopiert werden. In Zeile 5 wird das Datum wiederholt und über ein Spezialzahlenformat in den Wochentag umgewandelt.

B6: =DATUM(Jahr;\$B\$3;SPALTE()-1), Zahlenformat TT

B5: =B6, Zahlenformat TTT

Um die unterschiedliche Anzahl Tage pro Monat und den 29. Februar in Schaltjahren zu berücksichtigen, wird dieser über eine Formel berechnet, die den Folgetag des 28. Februar abprüft:

AD6: =WENN(TAG(\$AC\$6+1)=1;0;\$AC\$6+1)

Tragen Sie jetzt in Spalte A ab Zelle A7 die Namen der Arbeitnehmer ein. Die Spalten B:AF erhalten eine Spaltenbreite von 30 Pixel.

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q	R	S	T	U
1	Lohnabrechnung																				
2	Abrechnungsjahr:	2010																			
3	Abrechnungsmonat:	Januar																			▼
4																					
5	=B5	Fr	Sa	So	Mo	Di	Mi	Do	Fr	Sa	So	Mo	Di	Mi	Do	Fr	Sa	So	Mo	Di	Mi
6		01	02	03	04	05	06	07	08	09	10	11	12	13	14	15	16	17	18	19	20
7	Herbert Maier																				
8	Rudolf Spranger	=DATUM(Jahr,\$B\$3:SPALTE()-1)																			
9	Günther Hofbauer																				
10	Bernd Rösner																				
11	Erhan Öztürk																				
12	Walter Schmidt																				

Abbildung 3: Zwei Datumsreihen mit Verknüpfung zu Jahr und Monatsauswahl

Feiertage berechnen

Um die Eingabe und Berechnung der Arbeitsstunden zu erleichtern, bekommt der gesamte Tabellenbereich ein Bedingungsformat, das Samstage und Sonntage mit einem Hintergrundmuster versieht. Damit auch Feiertage und weitere freie Tage (z.B. Betriebsferien) berücksichtigt werden, stellt ein Tabellenblatt Feiertage die Liste der Feiertage zur Verfügung. Der Bereich mit den Datumswerten wird mit dem Bereichsnamen FLISTE versehen.

Tipp: Die Feiertagsberechnung ist für die einzelnen Bundesländer unterschiedlich. Laden Sie die Datei von der Webseite der Autoren (www.excellent-controlling.de im Menü *Excel-Toolbox*, *Feiertage Deutschland.xls* oder *Feiertage Österreich.xls*). Kopieren Sie die Tabelle in die Lohnabrechnung und weisen Sie den Datumswerten den Bereichsnamen FLISTE zu.

FTAGE		=DATUM(\$A\$1;1:1)*WERT(TEIL(\$D\$1;ZEILE();1))	
A	B	C	G
1	2010	Bundesland	Feiertage 2010 Bayern
2	<input type="radio"/>	Baden-Württemberg	Fr 01. Januar Neujahrstag
3	<input type="radio"/>	Bayern – überwiegend kath. Bevölkerung	Mi 05. Januar Hl. Drei Könige
4	<input checked="" type="radio"/>	Bayern	Fr 02. April Karfreitag
5	<input type="radio"/>	Bayern Stadtkreis Augsburg	Mo 05. April Ostermontag
6	<input type="radio"/>	Berlin	Sa 01. Mai Tag der Arbeit
7	<input type="radio"/>	Brandenburg	Do 13. Mai Christi Himmelfahrt
8	<input type="radio"/>	Mecklenburg-Vorpommern	Mo 24. Mai Pfingstmontag
9	<input type="radio"/>	Bremen	Do 03. Juni Fronleichnam
10	<input type="radio"/>	Hamburg	
11	<input type="radio"/>	Schleswig-Holstein	
12	<input type="radio"/>	Niedersachsen	So 03. Oktober Tag der d. Einheit
13	<input type="radio"/>	Hessen	
14	<input type="radio"/>	Nordrhein-Westfalen	Mo 01. November Allerheiligen
15	<input type="radio"/>	Rheinland-Pfalz	
16	<input type="radio"/>	Saarland	Sa 25. Dezember 1. Weihnachtsfeiertag
17	<input type="radio"/>	Saarland – überwiegend kath. Bevölkerung	So 26. Dezember 2. Weihnachtsfeiertag

Abbildung 4: Feiertagsberechnung je Bundesland

Bedingungsformat für Wochenenden und Feiertage

Excel bis Version 2003: *Format / Bedingte Formatierung*

Excel 2007/2010: Start / Formatvorlagen / Bedingte Formatierung.

Diese Regeln formatieren die arbeitsfreien Tage für die erste Zelle (B5):

Regel 1: =UND(WOCHENTAG(B\$5)=7;B\$5<>0)

Regel 2: =UND(WOCHENTAG(B\$5)=1;B\$5<>0)

Regel 3: =UND(NICHT(ISTNV(VERGLEICH(B\$5;FTAGE;0)));B\$5<>0)




Abbildung 5: Bedingungsformate für Wochenenden und Feiertage

Lohnabrechnung mit Matrixformeln

Für die Berechnung der Arbeitsstunden pro Monat und Arbeitnehmer reicht eine einfache Summe nicht aus. Die Arbeitsstunden müssen mit der Datumsreihe verglichen werden, damit die Abrechnung zwischen Wochenendstunden und Wochentagsstunden unterscheiden kann. Im Bereich ab der Zelle AH7 startet die Berechnung.

AH7							f_x {=SUMME(WENN(WOCHENTAG(\$B\$6:\$AF\$6)=7;\$B7:\$AF7))}						
AA	AB	AC	AD	AE	AF	AG	AH	AJ	AK	AL	AM	AN	
	Di	Mi	Do	Fr	Sa	So	Samstage		Sonntage		Wochentage		
	26	27	28	29	30	31	Stunden	Lohn	Stunden	Lohn	Stunden	Lohn	Gesamtlohn
7							0	- €	0	- €	0	- €	- €
8							0	- €	0	- €	0	- €	- €
9							0	- €	0	- €	0	- €	- €
10							0	- €	0	- €	0	- €	- €
11							0	- €	0	- €	0	- €	- €
12							0	- €	0	- €	0	- €	- €
13													

Abbildung 6: Matrixformel zur Berechnung der Wochenendstunden

Die erste Formel vergleicht die Arbeitsstunden des ersten Arbeitnehmers mit den Wochentagen der Datumsreihe B6:AF6. Die Kombination aus SUMME() und WENN() summiert nur Stunden auf, die der Bedingung entsprechen. Damit die Formel die Matrix Element für Element vergleichen kann, muss sie mit der Tastenkombination Strg + Umschalt + Eingabe abgeschickt werden.

AH7: =SUMME(WENN(WOCHENTAG(\$B\$6:\$AF\$6)=7;\$B7:\$AF7))

Tipp: Drücken Sie die Tastenkombination auch nach jeder Formeländerung. Geschweifte Klammern rund um die Formel weisen darauf hin, dass es sich um eine Matrixformel handelt.

Auch die Sonntagsstunden werden über eine Matrixformel über den Vergleich zwischen Datumsreihe und Arbeitnehmerstunden berechnet. Die Wochenstunden berechnen sich aus der Differenz zwischen der Stundensumme und den beiden Wochenendstundensummen.

$AJ7: =SUMME(WENN(WOCHENTAG(\$B\$6:\$AF\$6)=1; \$B7:\$AF7))$

$AL7: =SUMME(B7:AF7) - AH7 - AJ7$

Für die Berechnung der Lohnzahlungen wird die Stundensumme mit dem Grundlohn und den Zuschlägen summiert. Die Basisdaten enthalten die jeweiligen Eurobeträge, die mit passenden Bereichsnamen versehen sind.

$AI7: =AH7*Grundlohn+AH7*Grundlohn*ZSamstag$

$AK7: =AJ7*Grundlohn+AJ7*Grundlohn*ZSonntag$

$AM7: =AL7*Grundlohn$

$AN7: =AI7+AK7+AM7$

Vorlage kopieren und Stunden eintragen

Kopieren Sie die Vorlage, schalten Sie im Kombinationsfeld auf den Abrechnungsmonat um und tragen Sie die Arbeitsstunden der Mitarbeiter in die Tagesspalten ein. Die Matrixformeln berechnen automatisch die Summen der Samstags- und Sonntagsstunden und die daraus resultierenden Zuschlagsbeträge.

Tipp: Setzen Sie die Spalten B:AF und AH:AN mit der Gliederungsfunktion eine Ebene tiefer, dann können Sie wahlweise die Stundenübersicht oder die Auswertung ausblenden.

Excel bis Version 2003: *Daten / Gruppieren und Gliederung*

Excel 2007/2010: *Daten / Gliederung / Gruppieren*

		Lohnabrechnung																												Samstage		Sonntage								
		Mo	Di	Mi	Do	Fr	Sa	So	Mo	Di	Mi	Do	Fr	Sa	So	Mo	Di	Mi	Do	Fr	Sa	So	Mo	Di	Mi	Do	Fr	Sa	So	Mo	Di	Stunden	Lohn	Stunden	Lohn					
2	Abrechnungsjahr:	2011																																						
3	Abrechnungsmonat:	November																																						
7	Herbert Maier	0	0	0	0	0	0	0	4	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	16	600,00 €	24	1.200,00 €
8	Rudolf Springer	0	0	0	0	0	0	0	4	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	32	1.200,00 €	32	1.600,00 €
9	Günther Hofbauer	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	32	1.200,00 €	32	1.600,00 €
10	Bernd Rötner	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	16	600,00 €	24	1.200,00 €
11	Erian Özlük	0	0	0	0	0	0	0	7,5	7,5	7,5	7,5	7,5	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	23,5	861,25 €	31,5	1.575,00 €
12	Walter Schmidt	0	0	0	0	0	0	0	4	4	4	4	4	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	12	450,00 €	20	1.000,00 €	

Abbildung 7: Lohnabrechnung mit Matrixfunktionen und Gliederungsebenen

Praxisbeispiel 2: Arbeitsanfallanalyse

Die Personaleinsatzplanung hat vorrangig die Aufgabe, für die anfallende Arbeit genug Personal zur Verfügung zu stellen. Das erhält und erhöht die Produktivität, stellt die Kunden zufrieden und entlastet die Mitarbeiter. Um festzustellen, wie viel Arbeit in einem Unternehmensbereich vorhanden ist, muss der Arbeitsanfall der Vergangenheit analysiert werden. Arbeitsspitzen und Arbeitstäler, Über- und Unterbelastungen lassen sich aus diesen Erhebungen erkennen und verwerten.

Kriterien definieren

In der Arbeitsanfallanalyse werden zunächst die für das Unternehmen zutreffenden Kriterien definiert. Das sind in der Praxis Umsatz- oder Verkaufszahlen, geschriebene Aufträge, Kunden- oder Besucherfrequenzen, Telefonate, Supportleistungen usw. Fertigungsbetriebe definieren den Arbeitsanfall nach der Anzahl produzierter Artikel, Dienstleister und Programmierer rechnen Arbeitsstunden ab, Callcenter überwachen die Anzahl der Kundenkontakte. Im Tabellenblatt *Kriterien* ist eine Liste vorbereitet, die jederzeit abgeändert oder durch neue, eigene Kriterien ergänzt werden kann. Ein globaler Bereichsname auf der Liste sorgt dafür, dass die Liste in anderen Tabellenblättern zu verwenden ist.

	A	B
1	Kriterien	Alarm
2		Anmietung Fahrzeug
3		Aufträge
4		Besucherfrequenz
5		Kubikmeter
6		Kundenbesuche
7		Kundenfrequenz
8		Mietvertrag
9		Quadratmeter
10		Reinigungsarbeiten
11		Reklamation
12		Rückgabe Fahrzeug
13		Stückproduktion
14		Supportleistung
15		Telefonanrufe
16		Tickets
17		Überwachungseinsatz
18		Umsatz
19		Unfallhilfe
20		Verkauf
21		Versorgung Patienten

Abbildung 8: Kriterienliste mit Bereichsnamen Kriterien

Arbeitsanfall pro Woche

Das Tabellenblatt stellt die Kriterien aus der Kriterienliste in Gültigkeitsprüfungslisten zur Auswahl, im Beispiel sind drei Spalten mit den Kriterien Kundenfrequenz, Umsatz und Aufträge vorgesehen.

Excel bis Version 2003: *Daten / Gültigkeit*

Excel 2007/2010: *Daten / Datentools / Datenüberprüfung*

Zulassen: *Liste*

Quelle: *=Kriterien*

In der Zeile darunter wird ein Gewichtungsfaktor eingetragen und mit Bereichsnamen versehen. Bei wichtigen Kriterien ist dieser höher, bei weniger wichtigen geringer, die Summe muss immer 100 sein. Die Zellen sind mit einem Bedingungsformat versehen, das mit dieser Regel eine

Überschreitung überprüft und die Zellen entsprechend (rot) einfärbt:

Formelbedingung: $=\text{Gewicht1}+\text{Gewicht2}+\text{Gewicht3}>1$

	A	B	C	D	E	F	G
1	Arbeitsanfallanalyse						
2	Jahresverlauf						
3							
4		Kundenfrequenz		Umsatz		Aufträge	
5	Gewichtung	33%		33%			

Abbildung 9: Kriterienauswahl und Gewichtung

Die Daten (Stunden, Anzahl, Beträge) für die Kriterien trägt der Anwender zu jedem Wochentag in die gelb markierten Felder ein, auch die Anzahl der Arbeitsstunden muss erfasst werden. Die Formeln in Zeile 13 summieren die Einträge, und in den Spalten C, E und F werden die prozentualen Anteile berechnet. Der Arbeitsanteil berechnet sich aus den Produkten von Daten und Gewichtungen.

H6: $=C6*\text{Gewicht1}+E6*\text{Gewicht2}+G6*\text{Gewicht3}$

Für die Gegenüberstellung des gewichteten Arbeitsanfalls und der täglichen Arbeitsstunden sorgt ein Diagramm mit dem Diagrammtyp *Säulen-Flächen*. Die Arbeitsstunden aus Spalte J bilden die erste Datenreihe, der Arbeitsanfall steht in Flächenform als zweite Reihe im Diagramm.




Abbildung 10: Datenerfassung und Anteilberechnung für die einzelnen Wochentage

Arbeitsanfall im Jahresverlauf

Das Tabellenblatt für die Analyse des Jahresverlaufs ist ähnlich wie die Wochenübersicht aufgebaut, das Diagramm lässt sich aber mithilfe eines Formularelements auf das gesamte Jahresvolumen oder bis zu einem bestimmten Monat einstellen. Dazu wird das Kombinationsfeld mit einer Monatsnamenreihe und einer Verknüpfungszelle versehen. Die Bereiche, aus denen das Diagramm gebildet wird, berechnen sich über Formeln im Namens-Manager:

Rubrik:

```
=WENN(Jahresverlauf!$J$20=1;Jahresverlauf!$A$6:$A$17;BEREICH.VERSCHIEBEN(Jahresverlauf!$A$6;0;0;Jahresverlauf!$J$20-1;1))
```

Arbeitsanfall:

```
=WENN(Jahresverlauf!$J$20=1;Jahresverlauf!$H$6:$H$17;BEREICH.VERSCHIEBEN(Jahresverlauf!$H$6;0;0;Jahresverlauf!$J$20-1;1))
```

Arbeitsstunden:

```
=WENN(Jahresverlauf!$J$20=1;Jahresverlauf!$J$6:$J$17;BEREICH.VERSCHIEBEN(Jahresverlauf!$J$6;0;0;Jahresverlauf!$J$20-1;1))
```

Die Funktion DATENREIHE(), aus der das Diagramm gebildet wird, enthält diese Bereiche anstelle der absoluten Bezüge:

```
=DATENREIHE(Jahresverlauf!$H$4;Jahresverlauf!Rubrik;Jahresverlauf!Arbeitsanfall;2)
```




Abbildung 11: Arbeitsanfallanalyse im Jahresverlauf

Excel-Tipps für Personalcontroller

Tipp 1: Namen und Kostenstellen auffüllen

Viele Listen, die aus SAP oder anderen ERP-Systemen importiert werden, haben eine besonders nachteilige Aufteilung, die den Personalcontroller viel Zeit kostet: Mitarbeiternamen, Kostenstellen oder Kennziffern sind nur einmal pro Gruppe aufgeführt. Damit die Liste sinnvoll gefiltert, sortiert oder mit PivotTable-Berichten ausgewertet werden kann, müssen die Informationen umständlich und zeitaufwändig nachgebessert werden. Mit diesem Trick füllen Sie solche Gruppenelemente in wenigen Sekunden auf.

Das Beispiel zeigt eine Fehlzeitenanalyse, in der die Namen der Mitarbeiter und die Kostenstellen gruppiert sind.

	A	B	C	D	E
1		Fehlzeitenanalyse			
2		Einzelpersonenstatistik			
3					
4	Name	Abwesenheitsart/LFZ	Kostenstelle	Tage	Stunden
5	Karl Meier	0200 genommene Freischicht	40311	1,24	10,83
6		0250 Altersfreizeit		2,00	16,00
7		0300 Krankheit		16,25	134,00
8		0340 Arztbesuch		0,75	6,00
9		0360 Arbeits-/Wegeunfall		0,25	2,00
10		0380 Heilverfahren		4,00	32,00
11		0400 Jahresurlaub		4,00	32,00
12		0420 Freistellung Manteltarifvertrag		1,00	8,00
13		0790 Ausfallschicht Betriebsabteilung		1,00	8,00
14	Martin Schneider	0790 Ausfallschicht Betriebsabteilung	40311	1,00	8,00
15		0340 Arztbesuch		0,75	6,00
16		0380 Heilverfahren		6,00	48,00
17		0360 Arbeits-/Wegeunfall		0,25	2,00
18		0420 Freistellung Manteltarifvertrag		3,00	24,00
19		0400 Jahresurlaub		12,00	96,00
20		0200 genommene Freischicht		3,00	24,00
21		0300 Krankheit		20,00	160,00
22	Herbert Schilling	0790 Ausfallschicht Betriebsabteilung	40312	3,00	8,40
23		0340 Arztbesuch		0,50	4,50
24		0380 Heilverfahren		4,00	33,00
25		0360 Arbeits-/Wegeunfall		2,50	5,00

Abbildung 12: Fehlzeitenanalyse mit Gruppen für Namen und Kostenstellen

Fügen Sie eine neue, leere Spalte B ein und verknüpfen Sie die Überschrift und den ersten Mitarbeiternamen:

B4: =A4

B5: =A5

Schreiben Sie diese Funktion in die nächste Zelle der neuen Spalte. Sie prüft ab, ob der Name wechselt und trägt den neuen Namen ein. Ist die Nachbarzelle leer, wird der Name weiterverwendet:

=WENN(A6="" ;B5;A6)

Kopieren Sie die Formel per Doppelklick auf das Füllzeichen nach unten bis zum Ende der Liste. Die gleiche Formel füllt auch die Kostenstellen in einer neuen Spalte D.

	A	B	C	D	E	F	G
1			Fehlzeitenanalyse				
2			Einzelpersonenstatistik				
3							
4	Name	Name	Abwesenheitsart/LFZ	Kostenstelle	Kostenstelle	Tage	Stunden
5	Karl Meier	Karl Meier	0200 genommene Freischicht	40311	40311	1,24	10,83
6		Karl Meier	0250 Altersfreizeit	40311		2,00	16,00
7		Karl Meier	0300 Krankheit	40311		16,25	134,00
8			0340 Arztbesuch			0,75	6,00
9			0360 Arbeits-/Wegeunfall			0,25	2,00
10		Karl Meier	0380 Heilverfahren	40311		4,00	32,00
11		Karl Meier	0400 Jahresurlaub	40311		4,00	32,00
12		Karl Meier	0420 Freistellung Manteltarifvertrag	40311		1,00	8,00
13		Karl Meier	0790 Ausfallschicht Betriebsabteilung	40311		1,00	8,00
14	Martin Schneider	Martin Schneider	0790 Ausfallschicht Betriebsabteilung	40311	40311	1,00	8,00
15		Martin Schneider	0340 Arztbesuch	40311		0,75	6,00
16		Martin Schneider	0380 Heilverfahren	40311		6,00	48,00

Abbildung 12: Gruppeneinträge füllen mit WENN-Funktion

Jetzt müssen Sie nur noch die Formeln in Werte umwandeln und die „alten“ Spalten mit den

unvollständigen Namen bzw. Kostenstellen löschen. Markieren Sie dazu die Spalte B, kopieren Sie sie mit Strg + c in die Zwischenablage. Wählen Sie im Kontextmenü der rechten Maustaste *Inhalte einfügen / Werte*. Wandeln Sie so auch die Spalte D um und löschen Sie die Spalten A und E.

Tipp 2: Englisches Datum umwandeln

Das externe System liefert die Datumswerte einer Liste im US-Format YYYYMMDD, und das ist für Auswertungen nicht besonders nützlich. Greifen Sie zu Textfunktionen und holen Sie die einzelnen Datumselemente aus dem falschen Datum:

	A	B	C	D	E
1	Datum				
2	20100403				
3	20100404				
4	20100405				
5	20100406				
6	20100407				
7	20100408				
8	20100409				
9	20100410				
10	20100411				
11	20100412				

Abbildung 13: Datumswerte im US-Format, ...

B2: =LINKS (A2; 4)

C2: =TEIL (A2; 5; 2)

D2: =RECHTS (A2; 2)

Jetzt können Sie die Funktion DATUM() verwenden, um die Einzelteile wieder zusammenzufügen und ein echtes deutschsprachiges Datum zu erstellen:

E2: =DATUM (B2; C2; D2)

Natürlich brauchen Sie die Zwischenberechnung nicht, wenn Sie die Formeln gleich in die Datumsfunktion schachteln:

E2: =DATUM (LINKS (A2; 4) ; TEIL (A2; 5; 2) ; RECHTS (A2; 2))

E2		=DATUM(LINKS(A2;4);TEIL(A2;5;2);RECHTS(A2;2))					
	A	B	C	D	E	F	G
1	Datum	Jahr	Monat	Tag	Datum deutsch		
2	20100403	2010	04	03	03.04.2010		
3	20100404	2010	04	04	04.04.2010		
4	20100405	2010	04	05	05.04.2010		
5	20100406	2010	04	06	06.04.2010		
6	20100407	2010	04	07	07.04.2010		
7	20100408	2010	04	08	08.04.2010		
8	20100409	2010	04	09	09.04.2010		
9	20100410	2010	04	10	10.04.2010		
10	20100411	2010	04	11	11.04.2010		
11	20100412	2010	04	12	12.04.2010		

Abbildung 14: ... und mit Textfunktionen umgewandelt

Tipp 3: Alter berechnen

Für die Berechnung des Alters eines Mitarbeiters aus dem Geburtsdatum gibt es mehrere Varianten, von denen aber nur eine wirklich korrekt ist und das Alter stichtagsbezogen oder tagesgenau berechnet. Das Alter berechnet sich im einfachsten Fall über diese Formel, die das Geburtsjahr des Mitarbeiters (Zelle A1) vom aktuellen Jahr subtrahiert:

A1: 12.2.1980

B1: =JAHR(HEUTE())-JAHR(A2)

Diese Formel berücksichtigt aber nicht den Stichtag, in diesem Fall das Tagesdatum. Hatte der Mitarbeiter nämlich bereits Geburtstag in diesem Jahr, stimmt das Ergebnis um ein Jahr nicht. Für eine genauere Berechnung verwenden Sie eine WENN-Funktion, die prüft, ob das Geburtsdatum des Mitarbeiters vor oder nach dem Tagesdatum liegt:

=JAHR(HEUTE())-JAHR(A1)-WENN(DATUM(JAHR(HEUTE());MONAT(A1);TAG(A1))>HEUTE();1;0)

Alternative DATEDIF()

Einfacher berechnen Sie das Alter mit der Funktion DATDIF(). Geben Sie das Geburtsdatum, das Tagesdatum und „Y“ im letzten Argument an, und das Alter wird berechnet:

=DATEDIF(C6;HEUTE();"Y")

DATDIF() ist eine undokumentierte Funktion, die es schon sehr lange in Excel gibt. Sie berechnet Datumsdifferenzen, verwendet aber 31 Tage pro Monat zur Berechnung. Für exakte Analysen ist sie deshalb nicht geeignet.

Tipp 4: Kalenderwoche berechnen

Auch in den neuen Excel-Versionen 2007 und 2010 hat es Hersteller Microsoft leider nicht geschafft, Excel mit einer Funktion für die korrekte Berechnung der Kalenderwoche auszustatten. In Excel 2010 weist zumindest die Hilfefunktion darauf hin, dass die Woche, die als erste 4 Tage hat, die KW 1 ist (DIN 1330 oder ISO 8601), aber auch das zweite Argument der Funktion KALENDERWOCHE() löst das Problem nicht. Es regelt nur, an welchem Wochentag eine Woche beginnt:

=KALENDERWOCHE(Datum;Rückgabety))

Diese Formel ist zwar etwas voluminös, berechnet aber zuverlässig die KW eines Datums, und zwar auch, wenn dieses noch in die 52. oder 53. KW des Vorjahres fällt:

A1: 1.1.2011

A2: =KÜRZEN((A1-DATUM(JAHR(A1+3-REST(A1-2;7)));1;REST(A1-2;7)-9))/7)

Tipp 5: Negativzeiten berechnen

Um in Excel-Tabellen mit Zeiten rechnen zu können, sollten Sie sich entscheiden, ob die Werte mit oder ohne Datum zu berechnen sind. Muss die Tabelle Zeiträume über mehr als einen Tag berechnen, ist das Datum erforderlich, bei täglicher Arbeitszeiterfassung oder einfachen Stundenabrechnungen genügt die Zeitangabe, die rein technisch nur mit dem ersten Kalendertag rechnet. Eine einfache Arbeitszeitenberechnung könnte dann so aussehen:

	A	B	C	D
1	Name	Beginn	Ende	Arbeitszeit
2	Fritz Meier	08:00	16:00	08:00
3	Paul Huber	09:30	18:00	08:30
4	Beate Schneeberger	09:00	17:00	08:00
5	Dirk Rohrbach	07:00	12:00	05:00
6	Doris Wilhelms	07:30	14:00	06:30

Abbildung15: Berechnung von Zeitdifferenzen

Kritisch wird das Ganze bei Negativzeiten. Liegt das Arbeitsende vor dem Arbeitsbeginn (z. B. bei Schichtarbeit, Beginn 20:00 Uhr, Ende 4:00 Uhr), zeigt die Zelle mit der einfachen Formel „=Ende-Beginn“ eine durchlaufende Kette von #-Zeichen an. Der Grund ist einleuchtend, wenn das Prinzip der Zeitberechnung bekannt ist: Das Ergebnis würde im Bereich vor dem ersten Kalendertag landen, und für diesen lässt Excel keine Datums- oder Zeitberechnung zu. Eine (schlechte) Alternative ist die Umstellung auf das 1904-Datumsformat, das aus Kompatibilitätsgründen zum Apple Macintosh-System enthalten ist: Unter Extras / Optionen wird auf der Registerkarte Berechnen die Option 1904-Datumswerte angekreuzt. Damit ist der Kalender um 4 Jahre nach vorne versetzt, was für einfache Zeitberechnungen nützlich sein mag, die Tabellen aber gehörig durcheinanderbringt, sobald parallel mit Datums- und Zeitwerten gearbeitet wird. Die bessere Methode, um Negativzeiten zu berechnen, ist diese: Konstruieren Sie die Formel mithilfe der WENN-Funktion so, dass sie auch Zeitsprünge zwischen zwei Datumswerten richtig berechnet:

D2: =WENN (C2>B2;C2-B2;1-B2+C2)

Mit einer WENN-Formel lösen Sie auch das Problem der Sollzeitenberechnung, die zwar einfach als Differenz zwischen Arbeitszeit und Sollzeit errechnet wird, bei Zeitangaben, die unter dem Soll liegen, aber automatisch negativ und damit nicht darstellbar wird.

Tipp 6: Zeitwerte summieren

Bei der Summierung von Zeitwerten tritt häufig dieses ein Problem auf: Das Ergebnis beträgt nie mehr als 24 Stunden. Das hat einen einfachen Grund: Das Zahlenformat verhindert die Darstellung des Datums, es zeigt mit „hh:mm“ immer nur die Zeit des letzten Datumstags an. Liefert die Summe aber beispielsweise 38 Stunden, ist das für Excel der 1. Tag des Kalenders plus 14 Stunden. Die Zuweisung des Zahlenformats „TT.MM.JJ hh:mm“ würde es ans Licht bringen:

01.01.1900 14:00

Dieses Hindernis lässt sich natürlich umgehen. Weisen Sie der Zelle dieses Zahlenformat zu:

[hh]:mm

Die eckigen Klammern werden mit der *Alt-Gr*-Taste und 8 bzw. 9 erzeugt, sie sorgen dafür, dass die Gesamtstundenzahl und nicht das aus diesen resultierende Datum angezeigt wird.

XING-Gruppe “Excel im Controlling”

„Suchen Sie noch oder XINGen Sie schon?“ – Kommen Sie in unsere Gruppe. Hochkarätige Experten, interessante Kontakte, jede Menge nützliche Tipps und Tools für Controller erwarten Sie. Hier der Einladungslink:

<http://www.xing.com/group-51575.6443b9>



Kontakt Daten

EXCELLENTControlling

Angewandtes Controlling und Projektmanagement

Ignatz Schels und Prof. Dr. Uwe M. Seidel
Hemauerstraße 16
93047 Regensburg
www.excellent-controlling.de
info@excellent-controlling.de

Inhaltsverzeichnis

Impressum:

ControllerSpielwiese.de
Redaktion

St.-Martin-Str. 23, 68535 Edingen-Neckarhausen
E-Mail: Redaktion@ControllerSpielwiese.de
Internet: <http://www.controllerspielwiese.de>
Impressum und Geschäftsbedingungen

Feedback:

Anregungen und Feedback sind stets willkommen! Gerne hören wir von Ihnen: [Feedbackformular](#)